

MODELIZACIÓN DE LA DISTRIBUCIÓN DE ISÓTOPOS RADIACTIVOS EN EL ORGANISMO

Actualizado: 2017-06-12

Guillermo Sánchez. (http://diarium.usal.es/guillermo)

Notas:

La presentación está elaborada con el programa BIOKMOD. En la ayuda del programa están incluido la mayoria de los ejemplos.

Parte del material de este notebook procede del libro Mathematica Beyond Mathematics: The Wolfram Language in the Real World. Guillermo Sánchez. CRC Press. Disponible en AMAZON

Modelización compartimental (MC)

¿Dónde queremos llegar?

¿Cómo se modeliza un proceso biocinético/farmacocinético? :

Modelos lineales y no lineales

¿Cómo se pueden obtener los parámetros del modelo experimentalmente?

¿Qué es el diseño óptimo?

Modelización compartimental (MC)

Sistema físico o biológico que se descompone en un número finito de componentes llamados compartimentos que intercambian materia (partículas o flujo) entre ellos y/o con el exterior



Algunos usos

Modelización del metabolismo de la incorporación de partículas inhaladas, ingeridas injectadas.

Modelización de la incorporación por ingestión o inyección de compuestos a personas y otros seres vivos en Medicina y en Farmacia.

La MC se aplica en otras áreas como en el Transporte de partículas en estudios medioambientales

Ejemplo sencillos

La desintegración radiactiva: A -> B



$$\frac{dx(t)}{dt} = - k\, x(t)$$

Organo aislado

Supongamos un recipiente de volumen V lleno de agua con sal con un concentración q=o en t = o. Hay una entrada agua salada, con caudal v y concentración c, constantes. Se va produciendo instantáneamente una mezcla de concentración q(t) que va saliendo también a un caudal v.



$$q + dq = \frac{\overbrace{V\,q}^{\text{Sal que habia}} + \overbrace{c\,v\,dt}^{\text{Sal que entra}} - \overbrace{q\,v\,dt}^{\text{sal que sale}}}{V} \Rightarrow \frac{dq}{dt} = \frac{v(c-q)}{V}$$

Modelo bicompartimental simple

Consideremos el modelo de la figura



Formulación matematica

Llamamos $x_1(t)$ y $x_2(t)$, $t \geq 0$, a la cantidad de la especie retenida en los compartimentos 1 y 2 , respectivamente.

$$\frac{dx_1}{dt} = b(t) - \text{tasa trasferencias de } 1\, a\, 2$$

$$\frac{dx_2}{dt} = \text{flujo entrante desde } 1 - \text{salida hacia el exterior (elimination)}$$

Esta ecuacion se conoce **ecuación de balance de masas**. Si asumimos que la tasa de trasferencia, k_{12}, k_{20}, es proporcional a la masa (o concentracion) existente en el compartimento en t, entonces:

$$\frac{dx_1}{dt} = b(t) - k_{12}\, x_1$$

$$\frac{dx_2}{dt} = k_{12}\, x_1 - k_{20}\, x_2$$

En notación matricial:

$$\begin{pmatrix} x_1'(t) \\ x_2'(t) \end{pmatrix} = \begin{pmatrix} -k_{12} & 0 \\ k_{12} & k_{20} \end{pmatrix} + \begin{pmatrix} b(t) \\ 0 \end{pmatrix}$$

Modelo bicompartimental generalizado

Formulación matemática: Con entrada y salida al exterior desde los compartimentos 1 y 2



Llamamos $x_1(t)$ y $x_2(t)$ a las variables de estado (concentración, cantidad, etc) y su evolución en el tiempo, entonces el sistema podemos describirlo por el sistema de ecuaciones diferenciales siguientes

$$\frac{dx_1}{dt} = -\overbrace{(k_{12}+k_{10})}^{K_{12}} x_1 + k_{21}x_2 + b_1(t)$$

$$\frac{dx_2}{dt} = k_{12}x_1 - \overbrace{(k_{21}+k_{20})}^{K_{21}} x_2 + b_2(t)$$

Podemos reformularlo en notación matricial como sigue:

$$\mathbf{x}'(t) = \mathbf{A}\,\mathbf{x}(t) + \mathbf{b}(t)$$

$$\mathbf{x}'(t) = \begin{pmatrix} x'_1(t) \\ x'_2(t) \end{pmatrix} \quad \mathbf{A} = \begin{pmatrix} -K_{12} & k_{21} \\ k_{12} & -K_{21} \end{pmatrix} \quad \mathbf{x}(t) = \begin{pmatrix} x_1(t) \\ x_2(t) \end{pmatrix} \quad \mathbf{b}(t) = \begin{pmatrix} b_1(t) \\ b_2(t) \end{pmatrix}$$

Generalización a n compartimentos



$\dot{\mathbf{x}}(t) = \mathbf{A}\,\mathbf{x} + \mathbf{b}(t),\ t \geq 0$

$\mathbf{x}(0) = \mathbf{x}_0$

donde:

$x(t) = \{x_1(t),\ x_2(t),\ \ldots,\ x_n(t)\}^T$ siendo $x_i(t)$ la cantidad (masa, desintegraciones, concentración, etc) en el compartimento i en función de t.

A es una matriz $n \times n$ conocida como matriz compartimental

```
(para n = 3) A = ( a11  a12  a13 )
                 ( a21  a22  a23 );
                 ( a31  a32  a33 )
```

```
a11 = -(k10 + k12 + k13 + λ); a12 = k21; a13 = k31;
a21 = k12; a22 = -(k20 + k21 + k23 + λ); a23 = k32;
a31 = k13; a32 = k23; a33 = -(k30 + k31 + k32 + λ);
```

$\mathbf{b}(t) = \{b_1(t),\ b_2(t),\ \ldots,\ b_n(t)\}^T$ donde $\{b_i(t)\}$ es la entrada hacia el compartimento i desde el exterior del sistema

$\mathbf{x}(0) = \{x_1(0),\ x_2(0),\ \ldots,\ x_n(0)\}^T$ son las condiciones iniciales, esto es $x_i(0)$, representa la cantidad en el compartmento i en $t = 0$.

Ejemplo: Modelo del iodo.

Ejemplos: Resolución con BIOKMOD:

http://oed.usal.es/webMathematica/Biokmod/index.html



1 (blood), 2 (thyroid), 3 (rest of the body), 4 (blader), "0" ("out of the system") .
The rate transfer values (clearance), in days^{-1} (ICRP 78) are k10 = 1.9404, k12= 0.8316, k23 = Log[2]/80, k30 = 0.01155 and k31= 0.0462, k30 = 0.01155 and k40 = 12.
It assumed a continuouos input "1.2 Exp[-0.2 t]" in compartment 1
Also it is supossed as initial contition {1, 0, 0, 0}.
It is applied to I-131 Radioactive decay constant = Log[2]/8.0 (days^{-1}).

http://www3.enusa.es/webMathematica/Public/biokmod1.jsp

Solución

Enter the compartmental matrix:

```
{{1, 2, 0.83}, {1, 4, 1.94},
{2, 3, Log[2]/80}, {3, 0, 0.01155}, {3, 1, 0.0462}, {4, 0, 12}}
```

Number of compartments: 4 Decay constant: Log[2]/8 Initial conditions at time t = 0: {1,0,0,0}

Input function in each compartment : {1.2 Exp[-0.2 t],0, 0, 0}

All values of this field must be {0,.., 0} if it is an impulsive single-input, because the inputs are the initial conditions.

Time t to evaluate the content in each compartment (i.e.: t or {5, 20, 30}) : t

Range of t to be plotted: From t-min 0 to t-max 10

Time t, in days (kij mush be in days), to compute the acumulated disintegrations in each compartment : 50*365

Evaluate

Solution

$$x_1(t) \to 0.548312\, e^{-2.8566\, t} + 0.461614\, e^{-0.2\, t} - 0.0197996\, e^{-0.146776\, t} + 0.00987282\, e^{-0.0929699\, t}$$

$$x_2(t) \to -0.164814\, e^{-2.8566\, t} - 3.65968\, e^{-0.2\, t} + 0.3193\, e^{-0.146776\, t} + 3.50519\, e^{-0.0929699\, t}$$

$$x_3(t) \to 0.00052651\, e^{-2.8566\, t} + 0.570232\, e^{-0.2\, t} - 1.16135\, e^{-0.146776\, t} + 0.59059\, e^{-0.0929699\, t}$$

$$x_4(t) \to -0.188965\, e^{-12.0866\, t} + 0.115246\, e^{-2.8566\, t} + 0.0753393\, e^{-0.2\, t} - 0.00321706\, e^{-0.146776\, t} + 0.00159695\, e^{-0.0929699\, t}$$

Plot



Us (disintegrations in each compartment during a time t)

$$\{213\,522.,\ 1.85948 \times 10^{6},\ 111\,578.,\ 34\,271.9\}$$

Regresión. Ajuste de modelos

(Solo disponible en Biokmod)Usamos el paquete:

```
Needs["Biokmod`SysModel`"]
```

```
SysModel, version 1.5.1   2013-11-12
```

We have the model of the below figure. The rate transfer from compartment 2 to environment is known and its value is k_{20}= 0.05 d^{-1}.The experiment consists of an instantaneous injection b_1= 1 in compartment 1 and b_2= 0.5 in compartment 2 at t = 0. It supposed that in t = 0 the amount of substance in all compartment is "0".



The transfer coefficients k_{12} and k_{21} are unknown.

The amount in compartment 2 in different times (in days) was measured. Here are the data {t, x} (These data are been obtained by simulation)

```
list1 = {{0, 0.5}, {10, 0.30}, {20, 0.26}, {30, 0.24}, {40, 0.21},
   {50, 0.19}, {60, 0.17}, {70, 0.15}, {80, 0.135}, {90, 0.12}, {100, 0.11}};
```

We intend to estimate the parameters k_{12} and k_{21} by adjusting our model to the experimental data given in list1. We will proceed as follows

Step 1: The compartmental matrix of the model is defined as a function of parameters to be fitted

```
modelTwoCompart = CompartMatrix[2, {{1, 2, k12}, {2, 1, k21}, {2, 0, 0.05}}]
```

```
{{-k12, 0. + k21}, {0. + k12, -0.05 - k21}}
```

Step 2: The model is built by choosing the function of retention in compartment 2. In this case, how is a single-impulsive input, AcuteInput is used.

```
x2[t_, k12_, k21_] = x2[t] /. AcuteInput[modelTwoCompart, {1, 0.5}, t, x];
```

Step 3: Now it is fitted the coefficients k_{12} and k_{21} with the experimental data. Because `x2[t,k12,k21]` is an analytic expression NonlinearRegress can used.

We will need the *Mathematica* package NonlinearFit.

```
nlm = NonlinearModelFit[list1, x2[t, k12, k21], {{k12, 0.01, 0.5}, {k21, 1, 5}}, {t}]
```

```
FittedModel[ -0.230443 (0. + 1. e^(-0.363082 t)) + 0.230443 (≪1≫) + 0.5 (0.801291 (0. + 1. e^(-0.363082 t)) + 0.198709 (0. + 1. e^(-≪20≫ t))) ]
```

To get the functional form of the `FittedModel` object, use `Normal`:

```
Normal[nlm] // ExpandAll // Chop
```

$$0.170202\, e^{-0.363082\, t} + 0.329798\, e^{-0.0111678\, t}$$

The result is returned as a `FittedModel` object, of which properties can be returned:

```
nlm[{"ParameterTable", "ANOVATable"}]
```

	Estimate	Standard Error	t-Statistic	P-Value
k12	0.0810964	0.0122265	6.63285	0.0000955992
k21	0.243154	0.0422701	5.75238	0.000275471

	DF	SS	MS
Model	2	0.641483	0.320741
Error	9	0.0000420101	4.66779×10^{-6}
Uncorrected Total	11	0.641525	
Corrected Total	10	0.124414	

Here the fitted function and the experimental data are shown:

```
Plot[x2[t, 0.081, 0.2431], {t, 0, 100},
 Epilog→ {Hue[0], PointSize[0.02], Map[Point, list1]}]
```




```
Clear["Global`*"]
```

Diseño óptimo

```
Needs["Biokmod`Optdesign`"]
```

```
Optdesign, 1.0   2007-04-09
```

```
Optdes[ inp (e^(-2.0 t-0.09 p) + e^(-0.001 t-0.2 p)), t, {{inp, 100}, {p, 5}} , 0.5, 1, 2, 1]
```

```
{0.055815, {t0 → 0.5, t1 → 3.95733}}
```

Modelo dinámico

Let's consider the iodine biokinetic model before describe. The coefficient transfer values, in days^{-1}, taken from ICRP 78 are k_{10}= 1.9404, k_{30}= 0.01155 and k_{31}= 0.0462. We will suppose that k_{12} and k_{23} are unknown (ICRP gives for a standard man k_{12}= 0.8316, k_{23} = 0.0086625)

We will refer to iodine 131 which has a radioactive half-life of 8.02 days, this meaning that radioactive decay constant λ = ln 2/8.02 day^{-1}. Then the compartmental matrix is:

```
iodine131matrix = CompartMatrix[3, {{1, 2, k12}, {1, 0, 1.9404},
    {2, 3, k23} , {3, 0, 0.01155} , {3, 1, 0.0462}}, Log[2] / 8.02] // Chop
```

```
{{-2.02683 - k12, 0, 0.0462}, {k12, -0.0864273 - k23, 0}, {0, k23, -0.144177}}
```

A input b1= $27.13\, e^{-24.08\,t} + 27.13\, e^{-2.86\,t} - 0.02\, e^{-0.147\,t} + 0.0194\, e^{-0.093\,t}$ happens in compartment 1, and b=0 in the others (This kind of input happens in real situations when there is an input from the GIT (Gastro Intestinal) to the blood, for instance if the iodine is intaken by orally). Then:

```
binput = {-27.13 e^(-24.08 t) + 27.13 e^(-2.86 t) - 0.020 e^(-0.147 t) + 0.0194 e^(-0.093 t), 0, 0};
```

The initial condition are { 0, 0, 0}.

```
ic = {0, 0, 0};
```

The numerical solution of the system as function of the para metes {k12, k23} can be obtained using the package function ParametricSystemNDSolve,

```
{x1, x2, x3} = {x1 , x2, x3} /.
   ParametricSystemNDSolve[iodine131matrix, ic, binput, {t, 0, 100}, x, {k12, k23}];
```

Plot the solutions for different values of the parameter k12 and k23 .

```
Manipulate[
 LogLogPlot[{x1[k12, k23][t], x2[k12, k23][t] }, {t, 2, 100}, PlotLabels → {"x1", "x2"} ],
 {{k12, 0.83, "k13"}, 0.5, 1.2}, {{k23, 0.0086625, "k23"}, 0.001, 0.02}]
```



Modelos ICRP

Pueden resolverse con BiokmodWeb:

http://oed.usal.es/webMathematica/Biokmod/index.html ->ICRP Models ->Lung

http://oed.usal.es/webMathematica/Biokmod/index.html ->ICRP Models ->General

Ejemplo: Modelos no lineales

```
Quit[]
```

Ejemplo 1

Here is solved 2 D Fick' s law of diffusion from the boundaries of a circle

```
Ω = ImplicitRegion[(x^2 + y^2 ≤ 10), {{x, -5, 5}, {y, -5, 5}}];

eq1 = D[u[x, y, t], t] == 0.0000072 * (D[u[x, y, t], x, x] + D[u[x, y, t], y, y]) - 1.2;

sol = NDSolve[{eq1, DirichletCondition[u[x, y, t] == 100, x^2 + y^2 == 10],
    u[x, 0, t] == 10, u[0, y, t] == 10, u[x, y, 0] == 10}, u, {t, 0, 10}, {x, y} ∈ Ω];

Animate[ContourPlot[u[x, y, t] /. sol, {x, y} ∈ Ω, PlotRange → {0, 10}, ClippingStyle → Automatic,
  ColorFunction → "DarkRainbow", PlotLegends → Automatic], {t, 0, 100}]
```



Material adicional:
http://diarium.usal.es/guillermo

Parte del material de este notebook procede del libro Mathematica Beyond Mathematics: The Wolfram Language in the Real World. Guillermo Sánchez. CRC Press. Disponible en AMAZON

Tutoriales y presentaciones en youtube: http://diarium.usal.es/guillermo/mathematica/

BIOKMOD: http://diarium.usal.es/guillermo/biokmod/

Bibliografia: http://diarium.usal.es/guillermo/publicaciones/especializadas/

Sobre Biokmod, bioensayos y modelización compartimental

Sánchez G; "Fitting bioassay data and performing uncertainty analysis with BIOKMOD" Health Physics.. 92(1) :64-72. 2007. ISSN/ISBN: 0017-9078

Sánchez G; Biokmod: A Mathematica toolbox for modeling Biokinetic Systems". Mathematica in Education and Research: 10 (2) 2005. ISSN/ISBN: 1096-3324

Lopez-Fidalgo J; Sánchez G; Statistical Criteria to Establish Bioassay Programs. Health Physics.. 89 (4). 2005. ISSN/ISBN: 0017-9078

Sánchez G; Lopez-Fidalgo J "Mathematical Techniques for Solving Analytically Large Compartmental Systems" Health Physics..: 85 (2): 2003. ISSN/ISBN: 0017-9078

Sobre diseño óptimo:

Juan M. Rodríguez-Díaz;Guillermo Sánchez-León:"Design optimality for models defined by a system of ordinary differential equations" Biometrical Journal 56 (5), pag 886–900, September 2014

G. Sánchez; J. M. Rodríguez-Díaz . Optimal design and mathematical model applied to establish bioassay programs": Radiation Protection Dosimetry. doi:10.1093/rpd/ncl499. 2007. ISSN/ISBN: ISSN 1742-3406

Lopez-Fidalgo J Rodri1guez-Díaz J.M.,Sánchez G; G., Santos-Martín M.T. Optimal designs for compartmental models with correlated observations" Journal of Applied Statistic. 32, 2006 ISSN/ISBN: 0266-4763